# Maintainer: Maxime Gauduin <alucryd@archlinux.org>

pkgname=elementary-wallpapers
pkgver=7.0.0
pkgrel=1
pkgdesc='Collection of wallpapers for elementary OS'
arch=(any)
url=https://github.com/elementary/wallpapers
license=(GPL3)
groups=(pantheon)
makedepends=(
  git
  meson
)
_tag=fc16e4745dc429f48d7f45fe58a28f50c600a970
source=(elementary-wallpapers::git+https://github.com/elementary/wallpapers.git#tag=${_tag})
b2sums=(SKIP)

pkgver() {
  cd elementary-wallpapers
  git describe --tags
}

build() {
  arch-meson elementary-wallpapers build
  meson compile -C build
}

package() {
  DESTDIR="${pkgdir}" meson install -C build
}

# vim: ts=2 sw=2 et:
